>

UNAM
FACULTAD DE INGENIERÍA
DIVISIÓN DE CIENCIAS BÁSICAS
ECUACIONES DIFERENCIALES
GRUPO 11 SEMESTRE 2024-2
PRIMER EXAMEN PARCIAL Temas 1 & 2
SOLUCIÓN

2024-03-21

> *restart*

PREGUNTA 1 (20 puntos) Obtener la solución general de la siguiente ecuación diferencial ordinaria no lineal (***sin usar dsolve***)

> $Ecua := y' = \frac{2 \cdot x \cdot y}{3 \cdot x^2 - y^2}$

$$Ecua := \frac{d}{dx}\, y(x) = \frac{2\, x\, y(x)}{3\, x^2 - y(x)^2} \quad \textbf{(1)}$$

Respuesta

> $with(DEtools)$:

> $odeadvisor(Ecua)$

$$[[\_homogeneous,\ class\ A],\ \_rational,\ \_dAlembert] \quad \textbf{(2)}$$

> $EcuaBis := simplify(isolate(eval(subs(y(x) = x \cdot u(x), Ecua)), diff(u(x), x)))$

$$EcuaBis := \frac{d}{dx}\, u(x) = \frac{u(x)\left(-1 + \frac{2}{3 - u(x)^2}\right)}{x} \quad \textbf{(3)}$$

> $SolGral := int\left(\frac{1}{u\left(-1 + \frac{2}{3 - u^2}\right)}, u\right) - int\left(\frac{1}{x}, x\right) = \_C1$

$$SolGral := \ln(u+1) - 3\ln(u) + \ln(u-1) - \ln(x) = \_C1 \quad \textbf{(4)}$$

> $SolGralDos := simplify(\exp(lhs(SolGral))) = \_C1$

$$SolGralDos := \frac{u^2 - 1}{u^3\, x} = \_C1 \quad \textbf{(5)}$$

> $SolFinal := simplify\left(subs\left(u = \frac{y(x)}{x}, SolGralDos\right)\right)$

$$SolFinal := \frac{y(x)^2 - x^2}{y(x)^3} = \_C1 \quad \textbf{(6)}$$

> $DerSolFinal := isolate(diff(SolFinal, x), diff(y(x), x))$

$$DerSolFinal := \frac{d}{dx}\, y(x) = -\frac{2\, y(x)\, x}{y(x)^2 - 3\, x^2} \quad \textbf{(7)}$$

> $Ecua$

(8)

$$\frac{d}{dx}\, y(x) = \frac{2\,x\,y(x)}{3\,x^2 - y(x)^2} \quad (8)$$

> $Comprobar := simplify(rhs(DerSolFinal) - rhs(Ecua)) = 0$

$$Comprobar := 0 = 0 \quad (9)$$

Fin respuesta 1)

>

> restart

PREGUNTA 2 (20 puntos) Obtener la solución general de la siguiente ecuación diferencial ordinaria de coeficientes variables no homogenea (***sin usar dsolve***)

> $Ecua := x\cdot\log(x)\cdot y' - (1+\log(x))\cdot y + \frac{1}{2}\cdot\mathrm{sqrt}(x)\cdot(2+\log(x)) = 0$

$$Ecua := x\ln(x)\left(\frac{d}{dx}\, y(x)\right) - (1+\ln(x))\, y(x) + \frac{\sqrt{x}\,(2+\ln(x))}{2} = 0 \quad (10)$$

Respuesta

> $EcuaDos := lhs(Ecua) - \frac{\sqrt{x}\,(2+\ln(x))}{2} = rhs(Ecua) - \frac{\sqrt{x}\,(2+\ln(x))}{2}$

$$EcuaDos := x\ln(x)\left(\frac{d}{dx}\, y(x)\right) - (1+\ln(x))\, y(x) = -\frac{\sqrt{x}\,(2+\ln(x))}{2} \quad (11)$$

> $EcuaNorm := expand\left(\frac{lhs(EcuaDos)}{x\cdot\log(x)}\right) = expand\left(\frac{rhs(EcuaDos)}{x\cdot\log(x)}\right)$

$$EcuaNorm := \frac{d}{dx}\, y(x) - \frac{y(x)}{x\ln(x)} - \frac{y(x)}{x} = -\frac{1}{\sqrt{x}\,\ln(x)} - \frac{1}{2\sqrt{x}} \quad (12)$$

> $p := -\frac{1}{x\ln(x)} - \frac{1}{x}$

$$p := -\frac{1}{x\ln(x)} - \frac{1}{x} \quad (13)$$

> $q := rhs(EcuaNorm)$

$$q := -\frac{1}{\sqrt{x}\,\ln(x)} - \frac{1}{2\sqrt{x}} \quad (14)$$

> $IntPneg := simplify(\exp(-int(p, x)))$

$$IntPneg := x\ln(x) \quad (15)$$

> $IntPpos := simplify(\exp(int(p, x)))$

$$IntPpos := \frac{1}{x\ln(x)} \quad (16)$$

> $SolGral := y(x) = \_C1\cdot IntPneg + IntPneg\cdot int(IntPpos\cdot q, x)$

$$SolGral := y(x) = \_C1\, x\ln(x) + \sqrt{x} \quad (17)$$

> $Ecua$

$$x\ln(x)\left(\frac{d}{dx}\, y(x)\right) - (1+\ln(x))\, y(x) + \frac{\sqrt{x}\,(2+\ln(x))}{2} = 0 \quad (18)$$

> $Comprobar := simplify(eval(subs(y(x) = rhs(SolGral), Ecua)))$

$$Comprobar := 0 = 0 \quad (19)$$

Fin respuesta 2)

>

> *restart*

PREGUNTA 3 (30 puntos) Obtener la solución particular del siguiente problema de ecuaciones diferenciales ordinarias lineales no homogeneas con condiciones iniciales (***sin usar dsolve***)

> $Ecua := y'' - 6\cdot y' + 8\cdot y = 4\cdot \exp(2\,x) - 8\cdot \exp(4\,x)$

$$Ecua := \frac{d^2}{dx^2}\,y(x) - 6\,\frac{d}{dx}\,y(x) + 8\,y(x) = 4\,e^{2\,x} - 8\,e^{4\,x} \quad (20)$$

> $CondIni := y(0) = -4, D(y)(0) = 5$

$$CondIni := y(0) = -4, D(y)(0) = 5 \quad (21)$$

Respuesta

> $EcuaHom := lhs(Ecua) = 0$

$$EcuaHom := \frac{d^2}{dx^2}\,y(x) - 6\,\frac{d}{dx}\,y(x) + 8\,y(x) = 0 \quad (22)$$

> $Q := rhs(Ecua)$

$$Q := 4\,e^{2\,x} - 8\,e^{4\,x} \quad (23)$$

> $EcuaCarac := m^2 - 6\cdot m + 8 = 0$

$$EcuaCarac := m^2 - 6\,m + 8 = 0 \quad (24)$$

> $Raiz := solve(EcuaCarac)$

$$Raiz := 4, 2 \quad (25)$$

> $yy[1] := \exp(Raiz[1]\cdot x); yy[2] := \exp(Raiz[2]\cdot x)$

$$yy_1 := e^{4\,x}$$

$$yy_2 := e^{2\,x} \quad (26)$$

> $SolHom := y(x) = \_C1\cdot yy[1] + \_C2\cdot yy[2]$

$$SolHom := y(x) = \_C1\,e^{4\,x} + \_C2\,e^{2\,x} \quad (27)$$

> $SolNoHom := y(x) = AA\cdot yy[1] + BB\cdot yy[2]$

$$SolNoHom := y(x) = AA\,e^{4\,x} + BB\,e^{2\,x} \quad (28)$$

> $with(linalg) :$

> $WW := wronskian([yy[1], yy[2]], x)$

$$WW := \begin{bmatrix} e^{4\,x} & e^{2\,x} \\ 4\,e^{4\,x} & 2\,e^{2\,x} \end{bmatrix} \quad (29)$$

> $BB := array([0, Q])$

$$BB := \begin{bmatrix} 0 & 4\,e^{2\,x} - 8\,e^{4\,x} \end{bmatrix} \quad (30)$$

> $ParaVar := simplify(linsolve(WW, BB))$

$$ParaVar := \begin{bmatrix} -4 + 2\,e^{-2\,x} & 4\,e^{2\,x} - 2 \end{bmatrix} \quad (31)$$

> $AAprima := ParaVar[1]; BBprima := ParaVar[2]$

$$AAprima := -4 + 2\,e^{-2x}$$

$$BBprima := 4\,e^{2x} - 2 \quad (32)$$

> $AA := int(AAprima, x) + \_C1; BB := int(BBprima, x) + \_C2$

$$AA := -4\,x - e^{-2x} + \_C1$$

$$BB := -2\,x + 2\,e^{2x} + \_C2 \quad (33)$$

> $SolFinal := simplify(expand(SolNoHom))$

$$SolFinal := y(x) = (-2\,x + \_C2 - 1)\,e^{2x} - 4\,e^{4x}\left(x - \frac{\_C1}{4} - \frac{1}{2}\right) \quad (34)$$

> $SolFinalDos := y(x) = \_C10 \cdot \exp(4\,x) + \_C20 \cdot \exp(2\,x) - 4 \cdot x \cdot \exp(4\,x) - 2 \cdot x \cdot \exp(2\,x)$

$$SolFinalDos := y(x) = \_C10\,e^{4x} + \_C20\,e^{2x} - 4\,e^{4x}\,x - 2\,e^{2x}\,x \quad (35)$$

> $Comprobacion := simplify(eval(subs(y(x) = rhs(SolFinalDos), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprobacion := 0 = 0 \quad (36)$$

> $CondIni$

$$y(0) = -4, \mathrm{D}(y)(0) = 5 \quad (37)$$

> $EcuaUno := simplify(subs(x = 0, rhs(SolFinalDos) = -4))$

$$EcuaUno := \_C10 + \_C20 = -4 \quad (38)$$

> $EcuaDos := simplify(subs(x = 0, rhs(diff(SolFinalDos, x)) = 5))$

$$EcuaDos := 4\,\_C10 + 2\,\_C20 - 6 = 5 \quad (39)$$

> $Para := solve([EcuaUno, EcuaDos])$

$$Para := \left\{\_C10 = \frac{19}{2}, \_C20 = -\frac{27}{2}\right\} \quad (40)$$

> $SolPartDos := subs(Para, SolFinalDos)$

$$SolPartDos := y(x) = \frac{19\,e^{4x}}{2} - \frac{27\,e^{2x}}{2} - 4\,e^{4x}\,x - 2\,e^{2x}\,x \quad (41)$$

Fin respuesta 3)

>

> $restart$

PREGUNTA 4 (30 puntos) Obtener la solución general del siguiente problema de ecuaciones diferenciales ordinarias no homogeneas (***sin usar dsolve***)

> $Ecua := y'' + 4\,y = \sin(x) \cdot \sin(2\,x)$

$$Ecua := \frac{d^2}{dx^2}\,y(x) + 4\,y(x) = \sin(x)\,\sin(2\,x) \quad (42)$$

Respuesta

> $EcuaHom := lhs(Ecua) = 0$

$$EcuaHom := \frac{d^2}{dx^2}\,y(x) + 4\,y(x) = 0 \quad (43)$$

> $Q := rhs(Ecua)$

$$Q := \sin(x)\,\sin(2\,x) \quad (44)$$

> $EcuaCarac := m^2 + 4 = 0$

$$EcuaCarac := m^2 + 4 = 0 \tag{45}$$

> $Raiz := solve(EcuaCarac)$

$$Raiz := 2\,\mathrm{I}, -2\,\mathrm{I} \tag{46}$$

> $yy[1] := \cos(\mathrm{Im}(Raiz[1])\cdot x); yy[2] := \sin(\mathrm{Im}(Raiz[1])\cdot x)$

$$yy_1 := \cos(2\,x)$$

$$yy_2 := \sin(2\,x) \tag{47}$$

> $SolHom := y(x) = \_C1\cdot yy[1] + \_C2\cdot yy[2]$

$$SolHom := y(x) = \_C1\,\cos(2\,x) + \_C2\,\sin(2\,x) \tag{48}$$

> $SolNoHom := y(x) = AA\cdot yy[1] + BB\cdot yy[2]$

$$SolNoHom := y(x) = AA\,\cos(2\,x) + BB\,\sin(2\,x) \tag{49}$$

> $with(linalg) :$

> $WW := wronskian([yy[1], yy[2]], x)$

$$WW := \begin{bmatrix} \cos(2\,x) & \sin(2\,x) \\ -2\,\sin(2\,x) & 2\,\cos(2\,x) \end{bmatrix} \tag{50}$$

> $BB := array([0, Q])$

$$BB := \begin{bmatrix} 0 & \sin(x)\,\sin(2\,x) \end{bmatrix} \tag{51}$$

> $ParaVar := simplify(linsolve(WW, BB))$

$$ParaVar := \begin{bmatrix} -\frac{\sin(2\,x)^2\,\sin(x)}{2} & \frac{\cos(2\,x)\,\sin(x)\,\sin(2\,x)}{2} \end{bmatrix} \tag{52}$$

> $AAprima := ParaVar[1]; BBprima := ParaVar[2]$

$$AAprima := -\frac{\sin(2\,x)^2\,\sin(x)}{2}$$

$$BBprima := \frac{\cos(2\,x)\,\sin(x)\,\sin(2\,x)}{2} \tag{53}$$

> $AA := int(AAprima, x) + \_C1$

$$AA := \frac{\cos(x)}{4} + \frac{\cos(3\,x)}{24} - \frac{\cos(5\,x)}{40} + \_C1 \tag{54}$$

> $BB := int(BBprima, x) + \_C2$

$$BB := \frac{\sin(3\,x)}{24} - \frac{\sin(5\,x)}{40} + \_C2 \tag{55}$$

> $SolFinal := simplify(expand(SolNoHom))$

$$SolFinal := y(x) = 2\,\_C1\,\cos(x)^2 + 2\,\_C2\,\sin(x)\,\cos(x) + \frac{2\,\cos(x)^3}{5} - \_C1 - \frac{2\,\cos(x)}{15} \tag{56}$$

> $SolFinalDos := y(x) = \_C1\cdot\cos(2\,x) + \_C2\cdot\sin(2\,x) + \frac{2\cdot\cos(x)^3}{5} - \frac{2\cdot\cos(x)}{15}$

$$SolFinalDos := y(x) = \_C1\,\cos(2\,x) + \_C2\,\sin(2\,x) + \frac{2\,\cos(x)^3}{5} - \frac{2\,\cos(x)}{15} \tag{57}$$

> $Comprobacion := simplify(eval(subs(y(x) = rhs(SolFinalDos), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprobacion := 0 = 0 \quad \textbf{(58)}$$

Fin respuesta 4)

>

> *restart*

FIN DEL EXAMEN

>

>

>